

ФУНКЦИИ СУММИРОВАНИЯ И ПОДСЧЕТА

SUM (СУММ)

ПРИМЕРЫ ПРИМЕНЕНИЯ ФОРМУЛЫ SUM (СУММ)

У нас есть таблица с численностью населения в пяти городах РФ. Посчитаем с помощью SUM количество жителей этих городов.

Начнем с классического способа — выберем в формуле только те ячейки, которые нам нужно просуммировать:

<i>fx</i>	=СУММ(C2:C6)			
	A	B	C	D
1	№	Город	Численность населения	20 437 560
2		1 Москва	11 514 330	
3		2 Санкт-Петербур	4 848 742	
4		3 Новосибирск	1 473 737	
5		4 Екатеринбург	1 350 136	
6		5 Нижний Новгорс	1 250 615	
7				

Если захватить ячейки, содержащие текст, формула все равно будет работать:

<i>fx</i>	=СУММ(C1:C6)			
	A	B	C	D
1	№	Город	Численность населения	20 437 560
2		1 Москва	11 514 330	
3		2 Санкт-Петербур	4 848 742	
4		3 Новосибирск	1 473 737	
5		4 Екатеринбург	1 350 136	
6		5 Нижний Новгорс	1 250 615	

А еще можно выбрать сразу весь столбец, если мы уверены, что в нем только нужные нам данные и нет ничего лишнего:

fx | =СУММ(C:C)

	A	B	C	D
1	№	Город	Численность населения	20 437 560
2		1 Москва	11 514 330	
3		2 Санкт-Петербург	4 848 742	

(Чтобы все сработало, формулу в этом случае нужно вставлять в другой столбец, иначе она станет циклической и мы увидим ошибку.)

Также мы можем просуммировать отдельные ячейки (чтобы перейти к следующему элементу, выбирая ячейки или диапазоны мышкой, нажмите Ctrl):

fx | 14 238 682 x | Г | р. % .0_ .00 123 ~ | Arial ~ | 10 ~ | В

fx | =СУММ(C2;C4;C6)

	A	B	C	D
1	№	Город	Численность населения	=СУММ(C2;C4;C6)
2		1 Москва	11 514 330	
3		2 Санкт-Петербург	4 848 742	
4		3 Новосибирск	1 473 737	
5		4 Екатеринбург	1 350 136	
6		5 Нижний Новгород	1 250 615	

Или совместить выбор отдельных ячеек и диапазонов данных:

fx | 19 087 424 x | Г | р. % .0_ .00 123 ~ | Arial ~ | 10 ~ | В

fx | =СУММ(C2:C4;C6)

	A	B	C	D
1	№	Город	Численность населения	=СУММ(C2:C4;C6)
2		1 Москва	11 514 330	
3		2 Санкт-Петербург	4 848 742	
4		3 Новосибирск	1 473 737	
5		4 Екатеринбург	1 350 136	
6		5 Нижний Новгород	1 250 615	

КЕЙС «СОВМЕЩАЕМ В ОДНОЙ ЯЧЕЙКЕ ТЕКСТ И ФОРМУЛУ SUM (СУММ)»

В ячейке C1 находится заголовок таблицы, попробуем совместить его с формулой SUM:

<i>fx</i>	Численность населения		
	A	B	C
1	№	Город	Численность населения
2	1	Москва	11 514 330
3	2	Санкт-Петербург	4 848 742
4	3	Новосибирск	1 473 737
5	4	Екатеринбург	1 350 136
6	5	Нижний Новгород	1 250 615

Для этого обратим текст кавычками и добавим к нему формулу с помощью символа & (амперсанд):

<i>fx</i>	="Численность населения: "&СУММ(C2:C6)		
	A	B	C
1	№	Город	Численность населения: 20437560
2	1	Москва	11 514 330
3	2	Санкт-Петербург	4 848 742
4	3	Новосибирск	1 473 737
5	4	Екатеринбург	1 350 136
6	5	Нижний Новгород	1 250 615

Чтобы у числа появилась разрядность, добавим формулу ТЕКСТ с форматом "#,##0" (более подробно о форматах по ссылке: <https://support.google.com/docs/answer/3094139>):

<i>fx</i>	="Численность населения: "&ТЕКСТ(СУММ(C2:C6);"#,##0")			
	A	B	C	D
1	№	Город	Численность населения: 20 437 560	

А вот так это будет выглядеть, если использовать вместо амперсанда (&) формулу СЦЕПИТЬ (CONCATENATE):

<i>fx</i>	=СЦЕПИТЬ("Численность населения: ";ТЕКСТ(СУММ(C2:C6);"#,##0"))				
	A	B	C	D	
1	№	Город	Численность населения: 20 437 560		

Как вы могли понять, объединить можно все что угодно, например значения нескольких ячеек и текст:

<i>fx</i>	=C1&", "&D1&" и "&E1				
	C	D	E	F	G
1	Москва	Санкт-Петербург	Новосибирск	Екатеринбург	Нижний Новгород
2	11 514 330	4 848 742	1 473 737	1 350 136	1 250 615
3					
4	Москва, Санкт-Петербург и Новосибирск				

КЕЙС С ФУНКЦИЕЙ SUM. ФОРМУЛА

ОБРАЩАЕТСЯ ТОЛЬКО К ЗАПОЛНЕННОМУ УЧАСТКУ ДИАПАЗОНА

Вообще формула SUM очень простая — она суммирует все, что находится внутри ее скобок и что можно просуммировать: sum(a1:a). Но в нашей книге мы стараемся показывать интересные примеры, так что давайте я расскажу, как сделать так, чтобы формула обращалась только к заполненной части таблицы.

Метод, о котором пойдет речь далее, можно применять и в случае других, более ресурсоемких формул, чтобы не производить вычисления со столбцами целиком и не брать данные, выбирая диапазоны «с запасом».

У нас есть такая табличка:

f_x	A	B	C	D
1	Книга	Новинка?	Отгрузки	Тематика
2	100 имен любви	Новинка	634	Нон-фикшн
3	12 времён года	Новинка	628	Дизайн и творчество
4	365 дней очень творческого человека	Бэклист	526	Дизайн и творчество
5	365 дней очень творческого человека (бордовый)	Новинка	327	Дизайн и творчество
6	365 дней очень творческого человека (голубой)	Бэклист	405	Дизайн и творчество
7	365 дней очень творческого человека (малиновый)	Новинка	542	Дизайн и творчество
8	365 дней очень творческого человека (морской волны)	Бэклист	422	Дизайн и творчество
9	CD Rework (mp3)	Бэклист	943	Предприн-во и стартапы
10	CD Как привести дела в порядок	Бэклист	699	Личная эф-ть и психология
11	CD Камасутра для оратора	Новинка	910	Личная эф-ть и психология
12	CD Путь собственника	Новинка	921	Предприн-во и стартапы
13	CD Пятничные менеджер	Новинка	892	Нон-фикшн
14	CD Тайм-драйв	Бэклист	733	Личная эф-ть и психология
15	CD Уравнение счастья	Новинка	105	Личная эф-ть и психология
16	CD Харизма лидера диджипак	Бэклист	792	Личная эф-ть и психология
17				
18				

Сначала считаем количество занятых строк в столбце C с помощью COUNTA (СЧЁТЗ):

f_x	=СЧЁТЗ(C:C)					
1	A	B	C	D	E	F
1	Книга	Новинка?	Отгрузки	Тематика		
2	100 имен любви	Новинка	634	Нон-фикшн		16
3	12 времён года	Новинка	628	Дизайн и творчество		
4	365 дней очень творческого человека	Бэклист	526	Дизайн и творчество		
5	365 дней очень творческого человека (бордовый)	Новинка	327	Дизайн и творчество		
6	365 дней очень творческого человека (голубой)	Бэклист	405	Дизайн и творчество		
7	365 дней очень творческого человека (малиновый)	Новинка	542	Дизайн и творчество		
8	365 дней очень творческого человека (морской волны)	Бэклист	422	Дизайн и творчество		
9	CD Rework (mp3)	Бэклист	943	Предприн-во и стартапы		
10	CD Как привести дела в порядок	Бэклист	699	Личная эф-ть и психология		
11	CD Камасутра для оратора	Новинка	910	Личная эф-ть и психология		
12	CD Путь собственника	Новинка	921	Предприн-во и стартапы		
13	CD Пятничные менеджер	Новинка	892	Нон-фикшн		
14	CD Тайм-драйв	Бэклист	733	Личная эф-ть и психология		
15	CD Уравнение счастья	Новинка	105	Личная эф-ть и психология		
16	CD Харизма лидера диджипак	Бэклист	792	Личная эф-ть и психология		
17						
18						
19						
20						

Получается 16, и 16-я же последняя строка в нашем примере. Как вы уже, наверное, поняли, способ будет работать только в том случае, если данные занесены без пропусков строк.

Далее с помощью амперсанда (&) формируем наш диапазон:

= "C2:C" & СЧЁТЗ(C:C)

Вводим диапазон открытого типа с началом там, откуда начинается суммирование, и обрамляем его в кавычки: "C2:C". Далее, используя &, добавляем нашу формулу, которая вернет значения — количество строк, которые будут заполнены.

Результат формулы:

C2:C16 ×
= "C2:C" & СЧЁТЗ(C:C)

Добавляем формулу ДВССЫЛ (INDIRECT), чтобы другие формулы идентифицировали нашу конструкцию как диапазон ячеек:

=ДВССЫЛ("C2:C"&СЧЁТЗ(C:C))	
	628
	526
	327
	405
	542
	422
	943
	699
	910
	921
	892
	733
	105
	792

Обратите внимание, что выбранный диапазон сразу отображается в ячейках таблицы. Но нам нужно получить сумму его ячеек, а не выводить сам диапазон. Поэтому мы добавляем его в функцию СУММ (SUM):

=СУММ(ДВССЫЛ("C2:C"&СЧЁТЗ(C:C)))						
	A	B	C	D	E	F
	Книга	Новинка?	Отгрузки	Тематика		
	100 имен любви	Новинка	634	Нон-фикшн		9479
	12 времён года	Новинка	628	Дизайн и творчество		
	365 дней очень творческого человека	Бэклист	526	Дизайн и творчество		
	365 дней очень творческого человека (бордовый)	Новинка	327	Дизайн и творчество		
	365 дней очень творческого человека (голубой)	Бэклист	405	Дизайн и творчество		
	365 дней очень творческого человека (малиновый)	Новинка	542	Дизайн и творчество		
	365 дней очень творческого человека (морской волны)	Бэклист	422	Дизайн и творчество		
	CD Rework (mp3)	Бэклист	943	Предприн-во и стартапы		
	CD Как привести дела в порядок	Бэклист	699	Личная эф-ть и психология		
	CD Камасутра для оратора	Новинка	910	Личная эф-ть и психология		
	CD Путь собственника	Новинка	921	Предприн-во и стартапы		
	CD Пятничный менеджер	Новинка	892	Нон-фикшн		
	CD Тайм-драйв	Бэклист	733	Личная эф-ть и психология		
	CD Уравнение счастья	Новинка	105	Личная эф-ть и психология		
	CD Харизма лидера диджитал	Бэклист	792	Личная эф-ть и психология		

Получилось! Теперь, если пользователи будут обновлять табличку и заполнять новые строки в столбце С без пропусков, наша формула самостоятельно увеличит свои границы и правильно посчитает общую сумму.

SUMIF (СУММЕСЛИ)

КЕЙСЫ SUMIF С ЧИСЛОВЫМИ УСЛОВИЯМИ

С помощью функции SUMIF (СУММЕСЛИ) можно подсчитать сумму ячеек, которые соответствуют нашему условию.

Синтаксис функции:

=SUMIF(диапазон; условие; сумма диапазона)

=СУММЕСЛИ(диапазон; условие; сумма диапазона)

Например, мы можем получить общее количество жителей городов с численностью населения более 2 000 000 человек:

=СУММЕСЛИ(C2:G2;">2000000";C2:G2)							
	A	B	C	D	E	F	G
1			Москва	Санкт-Петербург	Новосибирск	Екатеринбург	Нижний Новгород
2	16 363 072		11 514 330	4 848 742	1 473 737	1 350 136	1 250 615

Если диапазоны условия и суммирования совпадают, то последний можно не указывать, ячейки будут просуммированы верно:

fx =СУММЕСЛИ(C2:G2; ">2000000")							
	A	B	C	D	E	F	G
1			Москва	Санкт-Петербург	Новосибирск	Екатеринбург	Нижний Новгород
2	16 363 072		11 514 330	4 848 742	1 473 737	1 350 136	1 250 615

Иногда может быть полезным не указывать условие в формуле, а сослаться на ячейку с ним, в нашем случае это ячейка B2:

fx =СУММЕСЛИ(C2:G2; B2)							
	A	B	C	D	E	F	G
1			Москва	Санкт-Петербург	Новосибирск	Екатеринбург	Нижний Новгород
2	16 363 072	>2000000	11 514 330	4 848 742	1 473 737	1 350 136	1 250 615

КЕЙСЫ SUMIF С ТЕКСТОВЫМИ УСЛОВИЯМИ

Условием SUMIF могут быть и текстовые значения. В примере ниже подсчитаем сумму операций с типом «зарплата»:

fx =СУММЕСЛИ(\$A\$2:\$A\$5; "зарплата"; \$B\$2:\$B\$5)			
	A	B	C
1	Тип	Сумма	400
2	зарплата	100	
3	зарплата	200	
4	аванс	50	
5	зарплата	100	

В текстовых условиях можно использовать символьные шаблоны. Давайте подсчитаем население городов, название которых начинается на "Во" (символ * обозначает любую последовательность символов):

fx =СУММЕСЛИ(B1:F1; "Во*"; B2:F2)						
	A	B	C	D	E	F
1		Барнаул	Владивосток	Волгоград	Воронеж	Екатеринбург
2	1 911 233	612 091	592 069	1 021 244	889 989	1 350 136

Чтобы заменить только один символ (допустим, мы не знаем, как правильно написать Барнаул), нужно использовать ?:

fx =СУММЕСЛИ(B1:F1;"Б?рнаул";B2:F2)						
	A	B	C	D	E	F
1		Барнаул	Владивосток	Волгоград	Воронеж	Екатеринбург
2	612 091	612 091	592 069	1 021 244	889 989	1 350 136

А под условие вида *а* попадут все города с одной или несколькими буквами а в названии.

fx =СУММЕСЛИ(B1:F1;"*а*";B2:F2)						
	A	B	C	D	E	F
1		Барнаул	Владивосток	Волгоград	Воронеж	Екатеринбург
2	3 575 540	612 091	592 069	1 021 244	889 989	1 350 136

Если добавить <> (не равно) в начало условия, то формула посчитает города без буквы а в названии:

fx =СУММЕСЛИ(B1:F1;"<>а*";B2:F2)						
	A	B	C	D	E	F
1		Барнаул	Владивосток	Волгоград	Воронеж	Екатеринбург
2	889 989	612 091	592 069	1 021 244	889 989	1 350 136

(<> можно использовать и в формулах SUMIF с числовыми условиями, например <>1 выберет все ячейки, не равные 1.)

SUMIFS (СУММЕСЛИМН)

КЕЙСЫ SUMIFS

С помощью формулы SUMIFS (СУММЕСЛИМН) можно подсчитать сумму ячеек, отвечающих нескольким условиям.

Посмотрим на таблицу ниже. Для начала найдем сумму всех зарплат сотрудника Васи Иванова:

	A	B	C	D
1	Дата	Сотрудник	Аванс / ЗП	Сумма
2	08.12.16	Вася Иванов	зп	1 000
3	15.12.16	Вася Иванов	аванс	500
4	15.12.16	Вася Иванов	зп	1 200
5	17.01.16	Слава Карелин	аванс	700
6	19.01.16	Слава Карелин	зп	2 500
7	23.01.16	Слава Карелин	аванс	1 000

Формула будет следующей (слева направо):

1-й элемент — диапазон суммирования;

2-й и 3-й — диапазон первого условия и само условие (в нашем случае столбец В «Сотрудник» и «Вася Иванов»);

3-й и 4-й — диапазон второго условия и само условие (обратите внимание, что формула нечувствительна к регистру: я написал **Зп**, а в нашей таблице встречаются только **зп**, однако формула все равно работает):

fx | =СУММЕСЛИМН(\$D\$1:\$D\$7;\$B\$1:\$B\$7;"Вася Иванов";\$C\$1:\$C\$7;"Зп")

	A	B	C	D	E	F
1	Дата	Сотрудник	Аванс / ЗП	Сумма		
2	08.12.16	Вася Иванов	зп	1 000		2 200
3	15.12.16	Вася Иванов	аванс	500		
4	15.12.16	Вася Иванов	зп	1 200		
5	17.01.16	Слава Карелин	аванс	700		
6	19.01.16	Слава Карелин	зп	2 500		
7	23.01.16	Слава Карелин	аванс	1 000		

Теперь подсчитаем общие выплаты сотрудникам по месяцам, чтобы понять, какие расходы понесла компания. Для этого нам нужно применить два условия к одному диапазону (столбец А с датами выплат):

- 1) больше или равно для первого числа месяца;
- 2) меньше или равно для последнего числа месяца.

Обратите внимание, что можно использовать ссылку на ячейку с датой, а можно прописать эту дату в формуле (в примере оба варианта):

fx	=СУММЕСЛИМН(\$D\$1:\$D\$7;\$A\$1:\$A\$7;">="&\$F2;\$A\$1:\$A\$7;"<="31.12.16")						
	A	B	C	D	E	F	G
1	Дата	Сотрудник	Аванс / ЗП	Сумма			
2	08.12.16	Вася Иванов	зп	1 000		01.12.16	2 700
3	15.12.16	Вася Иванов	аванс	500		01.01.17	4 200
4	15.12.16	Вася Иванов	зп	1 200			
5	17.01.17	Слава Карелин	аванс	700			
6	19.01.17	Слава Карелин	зп	2 500			
7	23.01.17	Слава Карелин	аванс	1 000			

Можно использовать формулу EOMONTH (КОНМЕСЯЦА) с аргументом 0, сославшись на ячейку с первым числом месяца, — она вернет последнее число этого месяца (эта формула подойдет для тех случаев, когда мы не знаем, сколько дней в месяце, или лень прописывать что-то руками):


fx	=СУММЕСЛИМН(\$D\$1:\$D\$7;\$A\$1:\$A\$7;">="&\$F2;\$A\$1:\$A\$7;"<="&КОНМЕСЯЦА(\$F2;0))						
-----------	---	--	--	--	--	--	--

Как и в формуле SUMIF, мы можем использовать символьные шаблоны. Давайте подсчитаем, сколько заработали все Васи за 2016 год (я знаю, что в таблице нет дат ранее 2016 года, поэтому буду использовать только одно условие на столбец с датой < 01.01.17):

f_x	=СУММЕСЛИМН(\$D\$1:\$D\$7;\$B\$1:\$B\$7;"Вася*";\$A\$1:\$A\$7;"<01.01.17")					
	A	B	C	D	E	F
1	Дата	Сотрудник	Аванс / ЗП	Сумма		
2	08.12.16	Вася Иванов	зп	1 000		2 700
3	15.12.16	Вася Иванов	аванс	500		
4	15.12.16	Вася Иванов	зп	1 200		
5	17.01.17	Слава Карелин	аванс	700		
6	19.01.17	Слава Карелин	зп	2 500		
7	23.01.17	Слава Карелин	аванс	1 000		

Кстати, иногда нам нужно, чтобы формула понимала символ * (или ?) как обычный текст. Чтобы этого добиться, нужно поставить перед ним тильду ~:

fx	=СУММЕСЛИМН(\$D\$1:\$D\$7;\$A\$1:\$A\$7;">=01.01.17";\$B\$1:\$B\$7;"~*")					
	A	B	C	D	E	F
1	Дата	Сотрудник	Аванс / ЗП	Сумма		
2	08.12.16	*	зп	1 000		3 200



[Почитать описание, рецензии
и купить на сайте](#)

Лучшие цитаты из книг, бесплатные главы и новинки:



[Mifbooks](#)



[Mifbooks](#)



[Mifbooks](#)